Differenciálegyenletes modellek

■ Inga

Tekintsük a következő egyparaméteres differenciálegyenletes modellt: $\phi' = \omega$, $\omega' = -g/l \sin(\phi)$, $l > 0$, $g = 9.81$. Keresd meg az egyensúlyi helyzetet. Oldd meg a rendszert az $\phi(0)$, $\omega(0)$ kezdeti adatokkal, ha $l_0 = 2$, $\phi(0) = 0$ és $\omega(0) = 3$.

Ábrázold a vektormezőt, a megoldás(oka)t és trajektóriát (statikus).
Készíts 2D-s interaktív ábrát, ahol az l paraméter
egy 1D-s csúszkán adott ([2,6]), s l aktuális értékének megfelelôen változik a vektormezô ill. a (0, 3)-bô l induló trajektória !

```
In[18]:= g = 9.81;
         sol =
          {ϕ[t], ω[t]} /. NDSolve[{ϕ'[t] == ω[t], ω'[t] == -g / l Sin[ϕ[t]], ϕ[0] == 0, ω[0] == 3} /. l → 2,
             {ϕ[t], ω[t]}, {t, 0, 5}][[1]]

Out[19]= {InterpolatingFunction[{{0., 5.}}, <>][t], InterpolatingFunction[{{0., 5.}}, <>][t]}

         Prepend[sol, t]

         {t, InterpolatingFunction[{{0., 5.}}, <>][t], InterpolatingFunction[{{0., 5.}}, <>][t]}

In[24]:= tr = ParametricPlot[sol, {t, 0, 5}, PlotStyle → {Red, Thickness[.01]}];

In[25]:= Show[VectorPlot[{ω, -g / l Sin[ϕ]} /. l → 2, {ϕ, -4, 4}, {ω, -4, 4}], tr]
```

Out[25]=



```
ParametricPlot3D[Evaluate[Prepend[sol, t]], {t, 0, 5}]
```




```
Plot[sol[[1]], {t, 0, 5}]
```



Iterációk, rekurziók (differenciaegyenletek)

```
Clear[f, F, x0, x]
```

```
Nest[F, x0, 3]
```

```
F[F[F[x0]]]
```

```
Nest[F, x0, 0]
```

```
x0
```

```
NestList[F, x0, 3]
```

```
{x0, F[x0], F[F[x0]], F[F[F[x0]]]}
```

```
f[x_] = x^2 - 1;
```

```
NestList[f, 1, 8]
```

```
{1, 0, -1, 0, -1, 0, -1, 0, -1}
```

```
RSolve[{y[n] == 2 y[n - 1], y[0] == 1}, y[n], n]
```

$\{\{y[n] \to 2^n\}\}$

```
RSolve[{y[n] == 2 y[n - 1]}, y[n], n]
```

$\{\{y[n] \to 2^{-1+n}\, C[1]\}\}$

Difference Systems

Feladat: Adjuk meg a rekurziót, vizsgáljuk meg a sorozatok elsô néhány tagját majd adjuk meg egy zárt alakot a megoldás sorozatokra

Adjunk meg: $\lim_\infty Q[n]/R[n]$

$$\mathrm{Q(n)}=\begin{pmatrix} 0 & \text{ha } n=0 \\ 2\,R(n-1)+1 & \text{ha } n>0 \end{pmatrix}$$

$$\mathrm{R(n)}=\begin{pmatrix} 0 & \text{ha } n=0 \\ Q(n)+Q(n-1)+1 & \text{ha } n>0 \end{pmatrix}$$

```
R[0] = 0;
Q[0] = 0;
Q[n_] := 2 R[n - 1] + 1;
R[n_] := Q[n] + Q[n - 1] + 1;
```

```
Clear[Q, R]
```

```
{Q[n], R[n]} /.
  RSolve[{Q[n] == 2 R[n - 1] + 1, R[n] == Q[n] + Q[n - 1] + 1, R[0] == 0, Q[0] == 0}, {Q[n], R[n]}, n][[
   1]] // FullSimplify
```

$$\left\{\frac{1}{6}\left(-6-\left(1-\sqrt{3}\right)^n\left(-3+\sqrt{3}\right)+\left(1+\sqrt{3}\right)^n\left(3+\sqrt{3}\right)\right),\ \frac{1}{6}\left(-6+\left(3-2\sqrt{3}\right)\left(1-\sqrt{3}\right)^n+\left(1+\sqrt{3}\right)^n\left(3+2\sqrt{3}\right)\right)\right\}$$

```
Q[n]
```

```
Table[Trace[Q[n]], {n, 0, 2}] // Simplify
```

```
{{{n, 0}, Q[0], 0}, {{n, 1}, Q[1], 2 R[1 - 1] + 1, {{{1 - 1, 0}, R[0], 0}, 2 × 0, 0}, 0 + 1, 1},
 {{n, 2}, Q[2], 2 R[2 - 1] + 1,
  {{{2 - 1, 1}, R[1], Q[1] + Q[1 - 1] + 1, {Q[1], 2 R[1 - 1] + 1, {{{1 - 1, 0}, R[0], 0}, 2 × 0, 0},
     0 + 1, 1}, {{1 - 1, 0}, Q[0], 0}, 1 + 0 + 1, 2}, 2 × 2, 4}, 4 + 1, 5}}
```

```
Table[%, {n, 8}] // Simplify
```

```
{{1, 2}, {5, 7}, {15, 21}, {43, 59}, {119, 163}, {327, 447}, {895, 1223}, {2447, 3343}}
```

Feladat.
Adjuk meg a köv inhom. egyenlethez tartozó magasabb hom egyenletet alapmegoldását, Mi a kapcsolat?
$a_n = a_{n-2} + 3$, $a_1 = 2$, $a_2 = 3$

```
Clear[α]
```

```
α[1] = 2; α[2] = 3;
α[n_] := α[n - 2] + 3
```

```
α[3]
```

```
5
```

```
Clear[a]
```

```
rule = a[n_] → a[n - 2] + 3
```

```
a[n_] → 3 + a[-2 + n]
```

```
a[n]
```

```
a[n]
```

```
(a[n + 1] /. rule) - (a[n] /. rule)
```

```
-a[-2 + n] + a[-1 + n]
```

```
Clear[b]
```

```
b[1] = 2;
b[2] = 3;
b[3] = 5;
b[n_] := b[n - 1] + b[n - 2] - b[n - 3];
```

```
b[4]
```

```
8
```

```
RSolve[{β[n] == β[n - 1] + β[n - 2] - β[n - 3], β[1] == 2,
  β[2] == 3,
  β[3] == 5}, β[n], n]
```

$$\left\{\left\{\beta[n] \to \frac{1}{4}\left(1 - (-1)^n + 6\,n\right)\right\}\right\}$$

```
Solve[x^3 - x^2 - x + 1 == 0, x]
```

```
{{x → -1}, {x → 1}, {x → 1}}
```

ℂ: Kitekintés: Newton fraktálok

```
NoOfIterations = 6; GridSize = .1;
```

```
F[z_] := z^3 - 2;
```

```
NS[z_] := z - F[z] / F'[z]
```

```
NestList[NS, 1 + I, 10] // N // Chop // TableForm
```

```
1. + 1. i
0.666667 + 0.333333 i
1.16444 - 0.737778 i
0.92614 - 0.17463 i
1.31644 + 0.156907 i
1.2463 + 0.0154548 i
1.25987 - 0.000338272 i
1.25992 + 2.60235 × 10^-8 i
1.25992
1.25992
1.25992
```

```
NestList[NS, I, 10] // N // Chop // TableForm
```

```
1. i
-0.666667 + 0.666667 i
-0.444444 + 1.19444 i
-0.606913 + 1.0646 i
-0.630766 + 1.09174 i
-0.629961 + 1.09112 i
-0.629961 + 1.09112 i
-0.629961 + 1.09112 i
-0.629961 + 1.09112 i
-0.629961 + 1.09112 i
-0.629961 + 1.09112 i
```

```
NestList[NS, -I, 10] // N // Chop // TableForm
```

```
-1. i
-0.666667 - 0.666667 i
-0.444444 - 1.19444 i
-0.606913 - 1.0646 i
-0.630766 - 1.09174 i
-0.629961 - 1.09112 i
-0.629961 - 1.09112 i
-0.629961 - 1.09112 i
-0.629961 - 1.09112 i
-0.629961 - 1.09112 i
-0.629961 - 1.09112 i
```

```
NSolve[F[z] == 0, z]
```

```
{{z → -0.629961 - 1.09112 i}, {z → -0.629961 + 1.09112 i}, {z → 1.25992}}
```

Három különbözô pontbólk indulva három kül. pontba tartanak a komplex iterációs sorozatok. Ezek éppen 3 (komplex) köbgyökei.

Konvergencia van 'majdnem' mindenhol, de hogy hová tartunk, nem nyilvánvaló: Ansatz, csak a 3 köbgyökhöz, ábrázoljuk egy medencét.
Az ábrázolás kényelemes a ListDensityPlot utasítással. Egy példa elôször: sakktábla

```
MatrixPlot[Table[If[EvenQ[i + j], 1, 0], {j, -2, 2}, {i, -2, 2}],
 Mesh → None, ColorRules → {1 → Black, 0 → White}]
```



```
z /. Solve[z^3 == 2, z]
```

$\{-(-2)^{1/3}, 2^{1/3}, (-1)^{2/3} 2^{1/3}\}$

```
Arg[%] // N
```

{-2.0944, 0., 2.0944}

Argumetumok szerint színezünk. Fehér pontok: a valós köbgyök a limesz.

```
Timing[MatrixPlot[Table[If[Abs[Arg[Nest[NS, i + j I, 6]] /. {Indeterminate → 10}] < 0.2, 1, 0],
   {j, -2., 2., GridSize}, {i, -2., 2., GridSize}], Mesh → None,
  Mesh → False, ImageSize → {500, 500}, ColorRules → {1 → Black, 0 → White}]]
```

{0.406,  }

3 köbgyök medencéje 3 kül. színnel.

```
Timing[MatrixPlot[Table[θ = Arg[Nest[NS, i + j I, NoOfIterations] /. {Indeterminate → 10}];
    If[θ > 1.8, 1, If[θ < -1.8, 2, 0]], {j, -2., 2., GridSize}, {i, -2., 2., GridSize}],
   Mesh → None, Mesh → False, ImageSize → {500, 500}, ColorRules → {2 → Red, 0 → Green, 1 → Blue}]]
```

{0.438,  }

Megjegyzés Konv. biz. - , érdekesebb nemkonvergens "attraktorok" -

Színezés konvergencia gyorsasága szerint is

```
GridSize = .01;

T = Table[θ = NestWhileList[NS, i + j I, Unequal, 2, 40] /. {Indeterminate → 10};
    {Last[Arg[θ]], Length[θ]}, {j, -2., 2., GridSize}, {i, -2., 2., GridSize}];

MM = Max[Flatten[T, 1][[All, 2]]]
```

41

```
T2 = Partition[
    Map[(If[#[[1]] < -1.8, 200, If[#[[1]] > 1.8, 0, 100]] + #[[2]]) &, Flatten[T, 1]], Length[T]];
```

```
MyCF2[n_] := If[n ≥ 200, RGBColor[1 - Mod[n, 200] / MM, 0, 0],
  If[n ≥ 100, RGBColor[0, 1 - Mod[n, 100] / MM, 0], RGBColor[0, 0, 1 - n / MM]]]

Timing[MatrixPlot[T2, Mesh → False, ImageSize → {500, 500},
  ColorFunction → MyCF2, ColorFunctionScaling → False]]
```

{3.672,

}

Ez eddig empirikus matematika. De a konvergencia bizonyítása lehetséges fixpont-té telek felhasználásával (vonzó/taszító fixpontok, deriváltak)

Fraktálalakzatok konstrukciója

Fák (Karsai János szívességéből)

Adott egy bázisobjektum (pl. egy lista, ami vektorokat tartalmaz).

Általános algoritmus:

Egy olyan szabállyal dolgozunk, ami egy lépésben kicserél egy objektumot egy másik/több másik objektumra; pl. egy szakaszt két szakaszra



Majd iteráljuk és vizualizájuk a folyamatot

Egy lehetséges bázisobjektum

```
In[1]:= object0 = {Line[{{0, 0}, {0, 1}}]};

In[2]:= object1 = {Line[{{-1, -1}, {1, 1}}], Line[{{1, 1}, {-1, -1}}],
           Line[{{-1, 1}, {1, -1}}], Line[{{1, -1}, {-1, 1}}]};

In[3]:= object2 = {Line[{{0, 0}, {0, 1}}], Line[{{1, 2}, {0, 2}}]};

        Graphics[object0]
```

Egyszerű szabály: Forgatás, zsugorítás, eltolás az eredeti végpontba

```
In[4]:= RotIm[point_List, ang_] := ( Cos[ang]  -Sin[ang] ).point
                                   ( Sin[ang]   Cos[ang] )

In[5]:= NewAutonomLine[obj_, α_, c_] := obj /. Line[u_] :→
           Sequence[
            Line[{u[[2]], u[[2]] + c RotIm[u[[2]] - u[[1]], α]}], Line[{u[[2]], u[[2]] + c RotIm[u[[2]] - u[[1]], -α]}]]

        NestList[NewAutonomLine[#, Pi / 4, .5] &, object0, 1]

        {{Line[{{0, 0}, {0, 1}}]},
         {Line[{{0, 1}, {-0.353553, 1.35355}}], Line[{{0, 1}, {0.353553, 1.35355}}]}}
```

```
Show[Graphics[NestList[NewAutonomLine[#, Pi / 4, .5] &, object0, 2]]]
```



```
Show[Graphics[object0], Graphics[NewAutonomLine[object0, Pi / 4, .5]]]
```



Feladat: iteráljuk cseréljük ki a bázisobjektumot, szöget stb. Használjuk a Manipulate/Listanimate utasítást, Ha kész vagyunk, színezzünk.

```
NestList[NewAutonomLine[#, Pi / 4, .5] &, object0, 1]

{{Line[{{0, 0}, {0, 1}}]},
 {Line[{{0, 1}, {-0.353553, 1.35355}}], Line[{{0, 1}, {0.353553, 1.35355}}]}}
```

In[7]:=
```
Manipulate[Show[Graphics[NestList[NewAutonomLine[#, Pi / 8, .8] &, object2, n]],
  PlotRange → {{-4, 4}, {-5, 5}}], {n, 0, 5, 1}]
```

Out[7]=




```
Clear[n]
```

Ugyanez színezzéssel (figyeljük meg az első argumentumot)

In[12]:=
```
NewAutonomLine[{n_?NumericQ, obj_}, α_, c_] := {n + 1, obj /. Line[u_] :→
    Sequence[
     Line[{u⟦2⟧, u⟦2⟧ + c RotIm[u⟦2⟧ - u⟦1⟧, α]}], Line[{u⟦2⟧, u⟦2⟧ + c RotIm[u⟦2⟧ - u⟦1⟧, -α]}]]}
```

```
NestList[NewAutonomLine[#, Pi / 4, .7] &, {0, object0}, 1]
```

```
{{0, {Line[{{0, 0}, {0, 1}}]}},
 {1, {Line[{{0, 1}, {-0.494975, 1.49497}}], Line[{{0, 1}, {0.494975, 1.49497}}]}}}
```

In[14]:=
```
TT = NestList[NewAutonomLine[#, Pi / 4, .7] &, {0, object1}, 5];
```

In[15]:=
```
TT2 = TT /. {n_?NumericQ, l_} → Graphics[{Hue[n / Length[TT]], l}];
```

In[16]:=
```
TT2[[2]] // InputForm
```

Out[16]//InputForm=
```
Graphics[{Hue[1/6], {Line[{{1, 1}, {1, 2.979898987322333}}], Line[{{1, 1}, {2.979898987322333
   Line[{{-1, -1}, {-1, -2.979898987322333}}], Line[{{-1, -1}, {-2.979898987322333, -1}}],
   Line[{{1, -1}, {2.979898987322333, -1}}], Line[{{1, -1}, {1, -2.979898987322333}}],
   Line[{{-1, 1}, {-2.979898987322333, 1}}], Line[{{-1, 1}, {-1, 2.979898987322333}}]}}]
```

In[17]:=
```
Manipulate[Show[Take[TT2, n], PlotRange → {{-5, 5}, {-5, 5}}], {{n, 4}, 1, 5, 1}]
```

Out[17]=



Koch pehely (segítség új szakaszt adjunk hozzá egy lépésben) HF

```
NewLines2[obj_, α_, c_] := obj /. Line[u_] :→ Sequence[Line[{u⟦1⟧, (2 u⟦1⟧ + u⟦2⟧) / 3}],
    Line[{(2 u⟦1⟧ + u⟦2⟧) / 3, (2 u⟦1⟧ + u⟦2⟧) / 3 + c RotIm[u⟦2⟧ - u⟦1⟧, α]}],
    Line[{(u⟦1⟧ + 2 u⟦2⟧) / 3 + c RotIm[u⟦2⟧ - u⟦1⟧, π - α], (u⟦1⟧ + 2 u⟦2⟧) / 3}],
    Line[{(u⟦1⟧ + 2 u⟦2⟧) / 3, u⟦2⟧}]]
```

```
object0 = {Line[{{0, 0}, {1, 0}}]}
```

```
{Line[{{0, 0}, {1, 0}}]}
```

```
object1 = {Line[{{-1, 0}, {0, 1}}]}
```

```
{Line[{{-1, 0}, {0, 1}}]}
```

```
NewLines2[object0, Pi / 3, 1 / 3]
```

$$\left\{\text{Line}\left[\left\{\{0, 0\}, \left\{\frac{1}{3}, 0\right\}\right\}\right], \text{Line}\left[\left\{\left\{\frac{1}{3}, 0\right\}, \left\{\frac{1}{2}, \frac{1}{2\sqrt{3}}\right\}\right\}\right],\right.$$
$$\left.\text{Line}\left[\left\{\left\{\frac{1}{2}, \frac{1}{2\sqrt{3}}\right\}, \left\{\frac{2}{3}, 0\right\}\right\}\right], \text{Line}\left[\left\{\left\{\frac{2}{3}, 0\right\}, \{1, 0\}\right\}\right]\right\}$$

```
Graphics[{object0, NewLines2[object0, Pi / 3, 1 / 3]}]
```



```
Graphics[NestList[NewLines2[#, Pi / 3, 1 / 3] &, object0, 4]]
```



```
Graphics[NestList[NewLines2[#, Pi / 2, 1 / 3] &, object0, 5]]
```



```
object1
```

```
{Line[{{-1, 0}, {0, 1}}]}
```

```
Graphics[NestList[NewLines2[#, Pi / 3, 1 / 3] &, object1, 4]]
```